Equilibria and Stability in the Management of a Renewable Resource

Michel De Lara and Luc Doyen

March 7, 2018

Contents

Let time t be measured in discrete units (such as years). Let $B(t)$ denote the biomass of a population at time t (beginning of time interval $[t, t+1[$). We consider the so called *Schaefer model*

$$B(t+1) = \mathsf{Biol}\big(B(t)\big) - h(t)\,, \quad 0 \leq h(t) \leq \mathsf{Biol}\big(B(t)\big) \tag{1}$$

where Biol is the population dynamics and $h(t)$ is the harvesting. Notice that, in the time interval $[t, t+1[$, growth of the stock occurs first, followed by harvesting[1].

The *sustainable yield* $h_e = \mathtt{Sust}(B_e)$ solves $B_e = \mathsf{Biol}(B_e) - h_e$, which gives:

$$\mathtt{Sust}(B) := \mathsf{Biol}(B) - B\,. \tag{2}$$

The *carrying capacity of the habitat* is the level $K > 0$ of positive biomass such that $\mathsf{Biol}(K) = K$, that is $\mathtt{Sust}(K) = 0$.

The *maximum sustainable yield* h_{MSE} and the corresponding *maximum sustainable equilibrium* B_{MSE} are

$$h_{\text{MSE}} := \sup_{B \geq 0}[\mathsf{Biol}(B) - B] \quad \text{and} \quad B_{\text{MSE}} := \arg\max_{B \geq 0}[\mathsf{Biol}(B) - B]\,. \tag{3}$$

From [1, p. 258] and numerical simulations, we shall consider the Pacific yellowfin tuna example as in Table 1.

[1] Another modelling would have been $B(t+1) = \mathsf{Biol}\big(B(t) - h(t)\big)$, $0 \leq h(t) \leq B(t)$.

	Pacific yellowfin tuna
yearly intrinsic growth	$R = 2.25$
carrying capacity	$K = 250\ 000$ metric tons
catchability	$q = 0.0\ 000\ 385$ per SFD
price	$p = 600$ \$ per metric ton
cost	$c = 2\ 500$ \$ per SFD

Table 1: Pacific yellowfin tuna data for a discrete time logistic model (adapted from [1, p. 258]). SFD: standard fishing day.

1 The Beverton-Holt model

The *Beverton-Holt model* is characterized by the discrete time dynamics mapping

$$\mathsf{Biol}(B) = \frac{RB}{1+bB}\,. \tag{4}$$

We have

$$h_{\text{MSE}} = \frac{(\sqrt{R}-1)^2}{b}\,, \quad B_{\text{MSE}} = \frac{\sqrt{R}-1}{b} \quad \text{and} \quad K = \frac{R-1}{b}\,. \tag{5}$$

Question 1 *Use the data in Table 1 to compute b in* (4)*. Give the maximum sustainable biomass B_{MSE} and the maximum sustainable yield h_{MSE} as in* (5)*.*

```
R_tuna = 2.25 ;
K_tuna = 250000 ; // metric tons

// BEVERTON-HOLT DYNAMICS
R_BH = R_tuna ;
b_BH = (R_BH-1) / K_tuna ;

function [y]=Beverton(B)
y=(R_BH*B)./(1 + b_BH*B) ;
y=maxi(0,y) ;
endfunction;

// SUSTAINABLE YIELD MACRO
function [SY]=sust_yield(dynamic,B), SY=dynamic(B)-B, endfunction;


// MAXIMUM SUSTAINABLE EQUILIBRIUM
B_MSE = (sqrt(R_BH) - 1)/b_BH ;
```

```
// maximum sustainable yield
h_MSE = sust_yield(Beverton,B_MSE) ;
// maximum sustainable yield
```

Question 2 *Select one biomass level B_e between the maximum sustainable biomass B_{MSE} and the carrying capacity K. Compute the corresponding sustainable yield h_e.*

Draw the corresponding steady trajectory of the Schaefer model (1) with the Beverton-Holt dynamics (4) *and* $h(t) = h_e$*. Pick up two different initial conditions in the neighborhood of the equilibrium biomass B_e. Draw the corresponding trajectories. Does the figure confirm or not the fact that the equilibrium biomass B_e is attractive?*

Recall that, for an equilibrium, being stable or attractive are unrelated notions.

Question 3 *Does the figure confirm or not the fact that the equilibrium biomass B_e is stable? Be specific in your justifications. What can you say about asymptotic stability of the equilibrium biomass B_e?*

```
// SUSTAINABLE EQUILIBRIUM

alpha=rand();
Be= alpha*B_MSE + (1-alpha)*K_tuna ;
// selection of one of many possible equilibria
he=sust_yield(Beverton,Be) ;
// corresponding sustainable yield


function [y]=sequential(y0,time,f)
[one,two]=size(Binit) ;
y=zeros(one,prod(size(time))) ; // time is a vector t0, t0+1,...,T
// vector will contain the trajectories y(1),...,y(T-t0+1)
// for different initial conditions
for k=1:one
y(k,1)=y0(k);
// initialization
for s=time(1:($-1)) -time(1)+1
// runs from 1 to T-t0+1
y(k,s+1)=f(s,y(k,s));
end ;
end ;
endfunction
```

```
// STATE TRAJECTORY UNDER DYNAMICS

function [y]=Beverton_e(t,B)
y=Beverton(B) - he ;
y=maxi(0,y) ;
endfunction
// Beverton-Holt dynamics with harvesting at equilibrium (Be,he)

T=20;
years=1:(T+1);

xset("window",31); xbasc(31);
Binit=Be;
Bt=sequential(Binit,years,Beverton_e);
plot2d2(years',Bt',1);
//
Binit=0.9*Be ;
Bt=sequential(Binit,years,Beverton_e);
plot2d2(years',Bt',2);
//
Binit=1.1*Be;
// It seems there is a bug with the previous version of 'sequential'
Bt=sequential(Binit,years,Beverton_e);
plot2d2(years',Bt',3);
//
xtitle('Trajectories with Beverton-Holt dynamics (R=' +string(R_tuna)...
+' and K=' +string(K_tuna) +')', 'years (t)','B(t)')
legends(['equilibrium biomass'],[1],'ur')
```

Question 4 *Find an equilibrium state B_e which is not attractive. Illustrate that B_e is not attractive with some trajectories. What can you say about asymptotic stability of the equilibrium biomass B_e?*

With price p, catchability coefficient q and harvesting unitary costs c, the *private property equilibrium* (PPE) is the equilibrium solution $(B_{\text{PPE}}, h_{\text{PPE}}) = (B_{\text{PPE}}, \mathtt{Sust}(B_{\text{PPE}}))$ which maximizes the rent as follows:

$$\max_{B \geq 0,\, h = \mathtt{Sust}(B)} [ph - \frac{ch}{qB}] \,. \tag{6}$$



Figure 1: Pacific yellowfin tuna biomass trajectories with Beverton-Holt dynamics

The *common property equilibrium* B_{CPE} makes the rent null and is given by

$$B_{\mathrm{CPE}} = \frac{c}{pq} \,. \tag{7}$$

Question 5 *Study the stability around the two following equilibria:*

- *common property equilibrium* B_{CPE}*,*
- *private property equilibrium*

$$B_{\mathrm{PPE}} = \frac{\sqrt{R(1+\frac{cb}{pq})} - 1}{b} \,. \tag{8}$$

Compare your observations with the theoretical results.

```
// Economic parameters
c_tuna=2500; // unit cost of effort
p_tuna=600; // market price
q_tuna=0.0000385; // catchability

c=c_tuna;
p=p_tuna;
q=q_tuna;
```

```
B_PPE= ( sqrt( R_BH * (1 + (b_BH*c/(p*q)) ) ) - 1 ) / b_BH;
// private property equilibrium

B_CPE=c/(p*q) ;
// common property equilibrium
```

2 The logistic model

The *logistic model* is characterized by the discrete time dynamics mapping

$$\mathsf{Biol}(B) = RB(1 - \frac{B}{\kappa}) \tag{9}$$

where $R \geq 1$ and $r = R - 1 \geq 0$ is the per capita rate of growth (for small populations), and κ is related to the carrying capacity K (which solves $\mathsf{Biol}(K) = K$) by:

$$\mathsf{Biol}(K) = K \iff RK(1 - \frac{K}{\kappa}) = K \iff \kappa = \frac{R}{R-1}K \iff K = \frac{R-1}{R}\kappa\,. \tag{10}$$

We have

$$h_{\mathrm{MSE}} = \frac{(R-1)^2}{4R}\kappa = \frac{R-1}{4}K \quad \text{and} \quad B_{\mathrm{MSE}} = \frac{R-1}{2R}\kappa = \frac{K}{2}\,. \tag{11}$$

Question 6 *Adapt the previous Scilab code to the logistic model, and compare the results.*

3 The Ricker model

The *Ricker model* is characterized by the discrete time dynamics mapping

$$\mathsf{Biol}(B) = B\exp(r(1 - \frac{B}{K}))\,. \tag{12}$$

Question 7 *Adapt the previous Scilab code to the Ricker model, and compare the results. Try numerical procedures: type help `fsolve` to obtain information about Scilab solver.*

References

[1] M. Kot. *Elements of Mathematical Ecology.* Cambridge University Press, Cambridge, 2001.